NAG Library Routine Document

F08UCF (DSBGVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08UCF (DSBGVD) computes all the eigenvalues and, optionally, the eigenvectors of a real generalized symmetric-definite banded eigenproblem, of the form

$$Az = \lambda Bz,$$

where A and B are symmetric and banded, and B is also positive definite. If eigenvectors are desired, it uses a divide-and-conquer algorithm.

2 Specification

```
SUBROUTINE F08UCF (JOBZ, UPLO, N, KA, KB, AB, LDAB, BB, LDBB, W, Z, LDZ,    &
                   WORK, LWORK, IWORK, LIWORK, INFO)
INTEGER            N, KA, KB, LDAB, LDBB, LDZ, LWORK,                       &
                   IWORK(max(1,LIWORK)), LIWORK, INFO
REAL (KIND=nag_wp) AB(LDAB,*), BB(LDBB,*), W(N), Z(LDZ,*),                  &
                   WORK(max(1,LWORK))
CHARACTER(1)       JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dsbgvd***.

3 Description

The generalized symmetric-definite band problem

$$Az = \lambda Bz$$

is first reduced to a standard band symmetric problem

$$Cx = \lambda x,$$

where C is a symmetric band matrix, using Wilkinson's modification to Crawford's algorithm (see Crawford (1973) and Wilkinson (1977)). The symmetric eigenvalue problem is then solved for the eigenvalues and the eigenvectors, if required, which are then backtransformed to the eigenvectors of the original problem.

The eigenvectors are normalized so that the matrix of eigenvectors, Z, satisfies

$$Z^{\mathrm{T}}AZ = \Lambda \quad \text{and} \quad Z^{\mathrm{T}}BZ = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Crawford C R (1973) Reduction of a band-symmetric generalized eigenvalue problem *Comm. ACM* **16** 41–44

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1977) Some recent advances in numerical linear algebra *The State of the Art in Numerical Analysis* (ed D A H Jacobs) Academic Press

5 Arguments

1: JOBZ – CHARACTER(1) *Input*

On entry: indicates whether eigenvectors are computed.

JOBZ = 'N'
Only eigenvalues are computed.

JOBZ = 'V'
Eigenvalues and eigenvectors are computed.

Constraint: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER(1) *Input*

On entry: if UPLO = 'U', the upper triangles of A and B are stored.

If UPLO = 'L', the lower triangles of A and B are stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

Constraint: N ≥ 0.

4: KA – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_a, of the matrix A.

If UPLO = 'L', the number of subdiagonals, k_a, of the matrix A.

Constraint: KA ≥ 0.

5: KB – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_b, of the matrix B.

If UPLO = 'L', the number of subdiagonals, k_b, of the matrix B.

Constraint: KA $\geq$ KB ≥ 0.

6: AB(LDAB, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \text{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix A.

The matrix is stored in rows 1 to $k_a + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB$(k_a + 1 + i - j, j)$ for $\max(1, j - k_a) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB$(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_a)$.

On exit: the contents of AB are overwritten.

7: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraint: $\text{LDAB} \geq \text{KA} + 1$.

8: $\text{BB}(\text{LDBB}, *)$ – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array BB must be at least $\max(1, \text{N})$.

On entry: the upper or lower triangle of the n by n symmetric band matrix B.

The matrix is stored in rows 1 to $k_b + 1$, more precisely,

if $\text{UPLO} = \text{'U'}$, the elements of the upper triangle of B within the band must be stored with element B_{ij} in $\text{BB}(k_b + 1 + i - j, j)$ for $\max(1, j - k_b) \leq i \leq j$;

if $\text{UPLO} = \text{'L'}$, the elements of the lower triangle of B within the band must be stored with element B_{ij} in $\text{BB}(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_b)$.

On exit: the factor S from the split Cholesky factorization $B = S^{\mathrm{T}} S$, as returned by F08UFF (DPBSTF).

9: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraint: $\text{LDBB} \geq \text{KB} + 1$.

10: W(N) – REAL (KIND=nag_wp) array *Output*

On exit: the eigenvalues in ascending order.

11: $\text{Z}(\text{LDZ}, *)$ – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \text{N})$ if $\text{JOBZ} = \text{'V'}$, and at least 1 otherwise.

On exit: if $\text{JOBZ} = \text{'V'}$, Z contains the matrix Z of eigenvectors, with the ith column of Z holding the eigenvector associated with $\text{W}(i)$. The eigenvectors are normalized so that $Z^{\mathrm{T}} B Z = I$.

If $\text{JOBZ} = \text{'N'}$, Z is not referenced.

12: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08UCF (DSBGVD) is called.

Constraints:

if $\text{JOBZ} = \text{'V'}$, $\text{LDZ} \geq \max(1, \text{N})$;
otherwise $\text{LDZ} \geq 1$.

13: $\text{WORK}(\max(1, \text{LWORK}))$ – REAL (KIND=nag_wp) array *Workspace*

On exit: if $\text{INFO} = 0$, $\text{WORK}(1)$ contains the minimum value of LWORK required for optimal performance.

14: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08UCF (DSBGVD) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the minimum sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if $N \leq 1$, LWORK ≥ 1;
if JOBZ = 'N' and $N > 1$, LWORK $\geq \max(1, 3 \times N)$;
if JOBZ = 'V' and $N > 1$, LWORK $\geq 1 + 5 \times N + 2 \times N^2$.

15: IWORK($\max(1, \text{LIWORK})$) – INTEGER array *Workspace*

On exit: if INFO $= 0$, IWORK(1) returns the minimum LIWORK.

16: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08UCF (DSBGVD) is called.

If LIWORK $= -1$, a workspace query is assumed; the routine only calculates the minimum sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if JOBZ = 'N' or $N \leq 1$, LIWORK ≥ 1;
if JOBZ = 'V' and $N > 1$, LIWORK $\geq 3 + 5 \times N$.

17: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$ and $i \leq N$, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

If INFO $= i$ and $i > N$, if INFO $= N + i$, for $1 \leq i \leq N$, then F08UFF (DPBSTF) returned INFO $= i$: B is not positive definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

8 Parallelism and Performance

F08UCF (DSBGVD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08UCF (DSBGVD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ = 'V' and, assuming that $n \gg k_a$, is approximately proportional to $n^2 k_a$ otherwise.

The complex analogue of this routine is F08UQF (ZHBGVD).

10 Example

This example finds all the eigenvalues of the generalized band symmetric eigenproblem $Az = \lambda Bz$, where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & 0 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ 0 & 0.63 & 0.48 & -0.03 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 2.07 & 0.95 & 0 & 0 \\ 0.95 & 1.69 & -0.29 & 0 \\ 0 & -0.29 & 0.65 & -0.33 \\ 0 & 0 & -0.33 & 1.17 \end{pmatrix}.$$

10.1 Program Text

```
    Program f08ucfe

!     F08UCF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dsbgvd, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: uplo = 'U'
!     .. Local Scalars ..
      Integer                          :: i, info, j, ka, kb, ldab, ldbb,      &
                                          liwork, lwork, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:), bb(:,:), w(:), work(:)
      Real (Kind=nag_wp)               :: dummy(1,1)
      Integer, Allocatable             :: iwork(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F08UCF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, ka, kb
      ldab = ka + 1
      ldbb = kb + 1
      liwork = 1
      lwork = 3*n
      Allocate (ab(ldab,n),bb(ldbb,n),w(n),work(lwork),iwork(liwork))

!     Read the upper or lower triangular parts of the matrices A and
!     B from data file

      If (uplo=='U') Then
        Read (nin,*)((ab(ka+1+i-j,j),j=i,min(n,i+ka)),i=1,n)
        Read (nin,*)((bb(kb+1+i-j,j),j=i,min(n,i+kb)),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ab(1+i-j,j),j=max(1,i-ka),i),i=1,n)
        Read (nin,*)((bb(1+i-j,j),j=max(1,i-kb),i),i=1,n)
```

```
      End If

!     Solve the generalized symmetric band eigenvalue problem
!     A*x = lambda*B*x

!     The NAG name equivalent of dsbgvd is f08ucf
      Call dsbgvd('No vectors',uplo,n,ka,kb,ab,ldab,bb,ldbb,w,dummy,1,work,    &
        lwork,iwork,liwork,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) w(1:n)
      Else If (info>n .And. info<=2*n) Then
        i = info - n
        Write (nout,99998) 'The leading minor of order ', i,                   &
          ' of B is not positive definite'
      Else
        Write (nout,99997) 'Failure in DSBGVD. INFO =', info
      End If

99999 Format (3X,(6F11.4))
99998 Format (1X,A,I4,A)
99997 Format (1X,A,I4)
    End Program f08ucfe
```

10.2 Program Data

```
F08UCF Example Program Data

  4      2      1             :Values of N, KA and KB

  0.24   0.39   0.42
        -0.11   0.79   0.63
               -0.25   0.48
                      -0.03   :End of matrix A
  2.07   0.95
         1.69  -0.29
                0.65  -0.33
                       1.17   :End of matrix B
```

10.3 Program Results

```
 F08UCF Example Program Results

 Eigenvalues
       -0.8305    -0.6401     0.0992     1.8525
```